ABSTRACT

In clinical trials, one of the most important safety reports created is the set of Adverse Event (AE) tables. One of the most common types of tables includes counts of event occurrences and individual subjects by Body System and Preferred Term. While the basic statistics presented in this type of table is the same (counts and percents), those tables are being produced by different subsets. Most commonly these tables are produced Overall and then by Seriousness, Fatality, Severity, Relationship to study drug, Outcome, and Action taken.

We present here a set of consistent macros that create all of the tables mentioned above with minimum effort on the part of the programmer. Simply by calling these macros and providing the standard variable names for Subject ID, Body System, Preferred Term, and any other related fields for which tables are required, a programmer can produce tables that can be used for both production and validation purposes.

KEYWORDS
Adverse Event, AE, Treatment Emergent, Body System, System Organ Class, Preferred Term, Serious, SAE, Severity, Relation, Outcome, Discontinue, Action.

AUDIENCE
Programmers, Statisticians and Database Professionals, with limited or advanced SAS® experience.

DEFINITION OF LIBRARIES

We will assume the following directory structure. Define the following libraries to read the input data and save the output. If the input data is in a different SAS version, use V8 or V6 in the library definition statement. If you do not specify the version, SAS reads datasets in any version, but for the output datasets it assumes the current version under which the program is executing.

\%LET root = %STR (c:\_research\ae) ;
LIBNAME sublib V9 "$root\data" ;
LIBNAME outlib V9 "$root\output" ;

MACRO COMPILING

To store a compiled macro, use the following statements before the macro definition. This will create the catalog file SASMACR.SAS7BCAT (under Microsoft Windows®), provided that you use the STORE option in the macro definition. In order to invoke the compiled macro, the same statements are used in the calling program to define its location. Using compiled macros preserves the generic code and eliminates the need for using %INC to include source code. The macro library MACLIB needs to be defined only once in a session and cannot be redefined or cleared until the end of the SAS session.

LIBNAME maclib V9 "$root\macros" ;
OPTIONS MSTORED SASMSTORE=maclib ;

INPUT DATA STRUCTURE

To avoid adding complexity to the main purpose of the macros, we assume that the structure of the data used for creating tables follows the Clinical Data Interchange Standards Consortium (CDISC) submission data standards for naming the dataset and variables. In CDISC, there are no user-defined Formats associated with variables and no Format Catalog is used. Instead, both a numeric version and a character description (decode) are contained in each data set. The numeric code (Variable Numeric Code, VARCD) is used for selection and ordering purposes and character decode (Variable Character Text Description, VAR) is used to explain the meaning of the numeric codes rather than using a format. In addition, we assume that Population Flags (Intent to Treat, Safety, Related …) have the character value “Y” for Yes or “N” for No.

Since real data does not always follow these guidelines, you can read the data in a work dataset, then rename the variables to their equivalent standard names. Depending on your data, you may have to save some variables in a different type (numeric or character). For the purpose of this paper, we will use ID for Unique Subject Number, TRT and TRTC for Treatment Group, BS for Body System, and PT for Preferred Term variable names. Our sample dataset structure is shown in Appendix 1.
BODY SYSTEM OR SYSTEM ORGAN CLASS

Some coding dictionaries use System Organ Class to indicate the Body System. For this reason we define the macro variable BSLBL, to define the label you want to see on the output table for body system.

%LET bslbl = %STR (System Organ Class) ;

PURPOSE OF PROGRAM

The macros presented here will produce a set of standard Adverse Events Tables that report frequency of occurrences as well as number and percent of patients with each event. The FINAL work dataset contains all the counts and percentages that will go in any table. For the final production table, we used ODS, PROC REPORT, and an RTF template, in order to have the table in an acceptable final layout. The FINAL dataset can also be used for validation. We will present here all the components that produce the tables. A sample of a final production table is shown in Appendix 2.

MACROS DESCRIPTION

(1) COUNTPOP macro

- Population or Randomization dataset count of unique subjects within each Treatment Group.
- Count the Number of Treatment Groups. Save it in a global macro variable.
- Subset the dataset for specific population.
- Output again for Total of all treatment groups. Update the number of treatment groups. Count Subjects in each group.
- Use %PUT to display macro variables results in the Log.

Input: DAT input dataset, including population flags.
POP population to select, such as Intent To Treat or Safety.

Output: NTRT global macro variable for Number of Treatment Groups.
TRTn, PCTn, HDRn (n=1, 2, ...) Counts, Percent and Header of each group.

%MACRO countpop ( dat = , pop = ) / STORE DES='Population Count' ;

**----- Count Number of Treatment Groups, Save it in a Macro Variable -----**;
PROC SQL ;
CREATE TABLE ntrt as
SELECT COUNT (DISTINCT trt) AS ntrt
FROM &dat ;
QUIT ;
DATA _NULL_ ;
SET ntrt ;
%GLOBAL ntrt ;
CALL SYMPUT ( 'ntrt' , PUT (ntrt,2.) ) ;
RUN ;

**----- Select Population, Output Again for Total of All Treatment Groups -----**;
%LET ntrt = %EVAL ( &ntrt + 1 ) ;
DATA dm0 (WHERE= (trt NE . AND &pop1='Y')) ;
SET &dat ;
OUTPUT ;
trt = &ntrt ;
trtc = 'Total' ;
OUTPUT ;
RUN ;

**----- Population Counts, Put Counts in Macro Variables -----**;
PROC SQL ;
CREATE TABLE pop as
SELECT COUNT (DISTINCT id) AS cnt, trt 
FROM   dm0 
GROUP BY trt ; 
QUIT ;
PROC TRANSPOSE PREFIX=trt OUT=pop (DROP=_NAME_) ; 
ID   trt ; 
VAR   cnt ; 
RUN ;
DATA pop ; 
SET ; 
ARRAY trt {*} 8. trt1-trt&ntrt ; 
ARRAY pct {*} 8. pct1-pct&ntrt ; 
DO i=1 TO &ntrt ; 
pct{i} = ROUND ( 100 * trt{i} / trt{i} , .1 ) ; 
END ; 
DROP   i ; 
RUN ;
DATA _NULL_ ; 
SET ; 
%DO i=1 %TO &ntrt ; 
%GLOBAL trt&i pct&i hdr&i ; 
CALL SYMPUT ( "trt&i" , PUT (trt&i,4.) ) ; 
CALL SYMPUT ( "pct&i" , PUT (pct&i,4.) ) ; 
CALL SYMPUT ( "hdr&i" , PUT (&i ,trt.) ) ; 
%END ; 
RUN ; 
%MEND countpop ;

(2) GETDATA macro
• Get AE Work dataset. Keep all records (AE Occurrences).
• Subset dataset for Population. Output again for Total of all treatment groups.
• For special tables where AEs occur more than 5% or twice as much as placebo, subset again for those preferred terms.

Input: DAT input work dataset, including all AE occurrences.
POP population to select, such as Intent To Treat or Safety.
Output: AE work dataset that has all records and a total subgroup.

%MACRO getdata ( dat = , pop = ) / STORE DES='Get Analysis Dataset' ;
DATA allr ;
SET &dat ;
IF aeterm NE ' ' AND &pop ;
  IF bs='' THEN bs='_NOT CODED' ;
  IF pt='' THEN pt='_NOT CODED' ;
  OUTPUT ;
  trt = &ntrt ;
  trtc = 'Total' ;
  OUTPUT ;
RUN ;
**----- Subset Data Again for Special Tables, if requested -----**;
%MEND getdata ;
(3) COUNTAE macro

- For Body System and Preferred Term – Counts of Records (Occurrences) and Subjects.
- First, we create 3 macro variables based on the group to run the counts for:
  - SRT is the sort by variable list (BS PT, BS, PT).
  - FRQ is the frequency list to used in PROC FREQ, including the asterisk "*" (ex: BS*PT*TRT).
  - FST is the condition to select the first record when counting distinct subjects (ex: if first.PT).
- The order (ORD) is used to order results in the table output. Usually, overall population counts come first, then those subjects who experienced adverse events, then classification by body system, then by preferred term.
- The order is also used to merge together the counts datasets of both Records and Subjects for the same category.

Input: DAT input dataset, including all AE occurrences.
   GRP group to run the counts and statistics for, Options:
   - BSPT for counting AEs per body system and preferred term combination,
   - BS for counting AEs per body system only,
   - PT for counting AEs per preferred term only,
   - SBJ for counting subjects who had an AE.
   TYP type of count, Options:
   - R for counting Records (Occurrences) of an AE,
   - S for counting Subjects who experienced an AE.
   ORD sort order in the final table output.

Output: COUNT&TYP.&ORD. work dataset that has the counts for a specific TYP (Records or Subjects).

The resulting dataset will be similar to:

<table>
<thead>
<tr>
<th>IDX</th>
<th>BS</th>
<th>PT</th>
<th>N1</th>
<th>N2</th>
<th>N3</th>
<th>N4</th>
</tr>
</thead>
<tbody>
<tr>
<td>4</td>
<td>Eye disorders</td>
<td>Dry eye . 3 3</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>4</td>
<td>Gastrointestinal disorders</td>
<td>Diarrhoea 3 5 7 15</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

%MACRO countae ( dat = , grp = , typ = , ord = ) / STORE DES='AE BS & PT Counts REC & SBJ' ;

**----- Create Macro Variables for Sorting, Frequency and Subject Selection -----**;
DATA _NULL_ ;
   %GLOBAL srt frq fst ;
   SELECT (UPCASE(COMPRESS("&grp"))) ;
      WHEN ("BSPT") DO  ;
         CALL SYMPUT ( "srt" , "bs pt" ) ;
         CALL SYMPUT ( "frq" , "bs*pt*trt" ) ;
         CALL SYMPUT ( "fst" , "if first.pt" ) ;
      END ;
      WHEN ("PT") DO  ;
         CALL SYMPUT ( "srt" , "pt" ) ;
         CALL SYMPUT ( "frq" , "pt*trt" ) ;
         CALL SYMPUT ( "fst" , "if first.pt" ) ;
      END ;
      WHEN ("BS") DO  ;
         CALL SYMPUT ( "srt" , "bs" ) ;
         CALL SYMPUT ( "frq" , "bs*trt" ) ;
         CALL SYMPUT ( "fst" , "if first.bs" ) ;
      END ;
      WHEN ("SBJ") DO  ;
         CALL SYMPUT ( "srt" , "" ) ;
         CALL SYMPUT ( "frq" , "trt" ) ;
         CALL SYMPUT ( "fst" , "if first.id" ) ;
      END ;
      OTHERWISE ;
      END ;
RUN ;
**----- Sort AE Dataset and Keep Only Required Fields -----**;
PROC SORT DATA=&dat OUT=bspt (KEEP=id trt bs pt);
   BY  trt id &srt; 
RUN ;

**----- If Subject Count, Select the First Record -----**;
DATA   bspt ;
   SET bspt ;
   BY  trt id &srt ;
   FORMAT bs pt $100. ;
   %IF %UPCASE(&typ)=S %THEN %DO ;
      &fst ;
   %END ;
RUN ;

**----- Count the Records, Sort to make it Ready for Transpose -----**;
PROC FREQ NOPRINT DATA=bspt ;
   TABLES &frq / OUT=count (DROP=PERCENT) ;
RUN ;

PROC SORT DATA=count OUT=count ;
   BY  
      %IF %UPCASE(&grp) NE SBJ %THEN %DO ;
      &srt ;
   %END ;
   trt ;
RUN ;

PROC TRANSPOSE PREFIX=n DATA=count OUT=count (DROP=_NAME_ _LABEL_) ;
   %IF %UPCASE(&grp) NE SBJ %THEN %DO ;
      BY  &srt ;
   %END ;
   ID  trt   ;
   VAR count ;
RUN ;

**----- Set Index, Used Later in Merge and Ordering Output -----**;
DATA   count&typ.&ord. ;
   SET count ;
   idx= &ord ;
RUN ;
%MEND countae ;

(4) REPORT macro

- ODS and REF Codes to produce an RTF type file.
- Report Definition: PROC REPORT for tabulation of the resulting dataset.
- You have the option of printing the AE Occurrence columns or not.

Input: DAT input dataset, which is the FINAL dataset including all the statistics.
Output: RTF AE table output.
%MACRO report (dat = ) / STORE DES='Report Definition' ;
    **----- RTF Code: Make Titles Bold & Indent PT -----**;
DATA final;
SET ;
LENGTH bspt $200 ;
IF SUBSTR(bs,1,1)='-' OR SUBSTR(pt,1,1)='-' THEN bspt = "^R\b" || TRANSLATE(bs,' ','-' );
ELSE bspt = "^R\li250 " || TRIM(LEFT(pt));
    IF bs = "-POPULATION DATASET" THEN bspt = "^R\b 'Population: &pop1" ;
    ELSE IF bs = "-ANY " || UPCASE("&bslbl") THEN bspt = "^R\b 'Any Adverse Events" ;
RUN ;
PROC SORT OUT=final ;
    BY idx0 bs idx DESCENDING n3 DESCENDING n2 DESCENDING n1 pt ;
RUN ;
**----- Create Table: RTF Setup & Assign Output File Name -----**;
OPTIONS NODATE NONUMBER PAPERSIZE=LETTER ORIENTATION=LANDSCAPE ;
ODS LISTING CLOSE ;
ODS ESCAPECHAR='^' ;
ODS RTF STYLE=TStyleRTF FILE="&root\output\&prgname.&outn..rtf" ;
**----- Report Titles & Footnotes -----**;
TITLE1 &title1 ;
TITLE2 &title2 ;
TITLE3 "&&&prgname.&outn.." ;
TITLE4 "&ttl4" ;
TITLE5 "Population: &pop1, &ttl5" ;
FOOTNOTE1 H=8PT J=L "^R\brdrt\brdrs\brdrw30 'Program: %UPCASE(&prgname..sas)" ;
PROC REPORT NOWD HEADSKIP CENTER MISSING SPLIT='|' SPACING=1 DATA=final ;
    COLUMN idx0 bs idx pt
        ( ("^R\brbrt\ql\bbslbl\|\b\" Preferred Term" bspt)
            %DO i=1 %TO &ntrt ;
                ("&&\hdr&i\n n (%)\ r&i npct&i)
            %END ;
        ) ;
    DEFINE idx0 / ORDER ORDER=DATA NOPRINT STYLE=[ASIS=ON] ;
    DEFINE bs / ORDER ORDER=DATA NOPRINT STYLE=[ASIS=ON] ;
    DEFINE idx / ORDER ORDER=DATA NOPRINT STYLE=[ASIS=ON] ;
    DEFINE pt / ORDER ORDER=DATA NOPRINT STYLE=[ASIS=ON] ;
    DEFINE bspt / DISPLAY LEFT STYLE=[CELLWIDTH=2.2IN ASIS=ON ] " FLOW ;
    %DO j=1 %TO &ntrt ;
        DEFINE r&j / DISPLAY RIGHT STYLE (COLUMN)=[CELLWIDTH=0.3IN] " ";
        DEFINE npct&j / DISPLAY RIGHT STYLE (COLUMN)=[CELLWIDTH=0.8IN] " ";
    %END ;
    COMPUTE AFTER bs ;
    LINE ' ' ;
    ENDCOMP ;
RUN ;
(5) PROCESS macro

- This is the main analysis loop macro that processes all data and produces one table at a time.
- Invoke COUNTPOP macro once to count the number of treatment groups and the main population groups.
- Invoke GETDATA followed by COUNTAE macro to count subjects and occurrences of each AE.
- The previous process is repeated 4 times for 4 Groups (BSPT, PT, BS, SBJ).
- For each group call the macros twice, once for counting Records and once for counting Subjects.
- Getting the dataset multiple times is not redundant. It avoids any previous subset and trimming of the data.
- The result is 8 datasets, 4 for records and 4 for subjects. We put all together.

Input:
- OUTN output File number, will be concatenated to program name.
- POP1 population to subset DM dataset for Subject count.
- POP2 population to subset AE dataset for Event count.
- TTL4 specific title of the table.
- TTL5 specific title of the table, including Population Subset.

Output: RTF FINAL dataset that goes into the AE table output.

%MACRO process ( outn= , pop1= , pop2= , ttl4= , ttl5= ) / STORE DES='AE Table Processing' ;

**----- COUNTPOP, GETDATA, and COUNTAE Macro Invocation -----**;
%countpop(dat= dm , pop= &pop1) ;
%getdata (dat= ae , pop= &pop2) ;
%countae (dat= allr , grp= bspt , typ= r , ord= 4 ) ;
%getdata (dat= ae , pop= &pop2) ;
%countae (dat= allr , grp= bspt , typ= s , ord= 4 ) ;
** Similar Calls for BS, PT, and SBJ, giving them different ORD **;

**----- Put All REC Count Together & All SBJ Count Together -----**;
DATA finalr (DROP= i n1-n&ntrt) ;
    SET countr1 countr2 countr4 ;
    ARRAY n{*} 8. n1-n&ntrt ;
    ARRAY r{*} 8. r1-r&ntrt ;
    DO i=1 to &ntrt ;
        r{i} = n{i} ;
    END ;
RUN ;
PROC SORT OUT=finalr ;
    BY idx bs pt ;
RUN ;
DATA finals ;
    SET counts1 counts2 counts4 ;
RUN ;
PROC SORT OUT=finals ;
    BY idx bs pt ;
RUN ;
**----- Statistics, Put Both SBJ & OCC Counts Together ------**
DATA final;
   MERGE finals finalr;
   BY idx bs pt;
RUN;

**----- Add Population Top Line ------**
DATA final;
   SET final END=eof;
   OUTPUT;
   IF eof THEN DO;
      idx = 0;
      bs  = "-POPULATION DATASET";
      pt  = "-ANY PREFERRED TERM";
      %DO i=1 %TO &ntrt;
         n&i = %EVAL (&&trt&i);
         r&i = . ;
      %END;
      OUTPUT;
   END;
RUN;

**----- Statistics: Put both SBJ & REC counts together ------**
DATA final (DROP= i trttot);
   SET ;
   FORMAT npct1-npct&ntrt $12. ;
   ARRAY n{*} 8. n1- n&ntrt ;
   ARRAY pct{*} $8. pct1- pct&ntrt ;
   ARRAY npct{*} $12. npct1-npct&ntrt ;
   DO i=1 TO &ntrt ;
      SELECT (i);
      WHEN (1)  trttot = &trt1 ;
      WHEN (2)  trttot = &trt2 ;
      WHEN (3)  trttot = &trt3 ;
      WHEN (4)  trttot = &trt4 ;
      OTHERWISE ;
   END ;
   ** Calculate PCT and Convert to Character for Report with Decimal Aligning **;
   IF n{1}^=. AND trttot>0 THEN pct{1} = PUT(ROUND(100*n{1}/trttot, .1), 5.1) ;
   ** Combine Count and Percent - Format to Align Both within One Variable **;
   IF n{1} NE . THEN DO ;
      IF pct{1} NE '100.0' THEN npct{1} = PUT(n{1},3.) || ' (' || COMPRESS(pct{1}) || ')' ;
      ELSE npct{1} = PUT(n{1},3.) || ' (' || COMPRESS(pct{1}) || ')' ;
   END ;
   ELSE npct{1} = ' 0' ;
   END ;
IF bs = "-POPULATION DATASET" THEN idx0 = 0 ;
ELSE IF bs = "-ANY "||UPCASE ("&bslbl") THEN idx0 = 1 ;
ELSE                                         idx0 = 2 ;
RUN ;

**----- Invoke Report Macro ------**;
%report (dat=final) ;
%MEND process ;

(6) The Calling Program

- Set up macro variables, define libraries, and system options.
- Bring in the population dataset, rename variables to standard naming if necessary.
- Bring in the AE dataset, rename variables to standard naming if necessary.
- Expected variable names in the datasets are: ID for subject number, BS for body system, PT for preferred term, TRT and TRTC for treatment group. Keep only required variables.
- For variables with missing values, consider the worst case scenario (ex: missing Severity is considered Severe).

**----- Formats Used ------**;
PROC FORMAT ;
  VALUE trt  1 = "Placebo"
          2 = "Drug A"
          3 = "Drug B"
          4 = "Total" ;
RUN ;

**------ SETUP Used ------**;
%LET prgname = TAE ;
%LET root    = %STR(c:\_Research\AE) ; ** Main Directory Path **;
%LET bslbl   = %STR(System Organ Class) ; ** Body System Label **;
LIBNAME sublib "&root\data" ;     ** Submission Data Library **;
LIBNAME maclib "&root\macros" ;     ** Complied Macros Library **;
OPTIONS NOCENTER MSTORED SASMSTORE=maclib ; ** To Save and Invoke Compiled Macros **;
ODS PATH maclib.templat(READ) SASHELP.tmplmst(READ) ; ** Define RTF Template Path **;

**------ Tables Number in Title ------**;
%LET tae1 = Table 5.1 ;  ** AEs, By BS & PT, All Reported Adverse Events **;

**------ Main Titles in All Tables ------**;
%LET title1= %STR(J=L "XYZ Pharmaceuticals"  J=R "Page ^{pageof}") ;
%LET title2= %STR(J=L "Protocol: ABC-123"  J=R "&sysdate9") ;

**------ FIX DATA: Get DM & AE Analysis Datasets, Rename Variables to Standard Names ------**;
**------ Keep Required Fields Only, Keep All Records ------**;

DATA dm (KEEP= id trt trtc safety complt) ;
  SET sublib.dm (RENAME=(usubjid=id armcd=trt arm=trtc)) ;
RUN ;

DATA ae (KEEP= id trtc safety aetext aeterm bs pt
  aereid aerecat aasevcat aasev aaser aeae aedth ) ;
  SET sublib.ae (RENAME=(usubjid=id armcd=trt arm=trtc aebodsys=bs aedecod=pt)) ;

**------ For Missing Data: Consider worst case scenario ------**;
**------ Make any necessary changes to required variables ------**;
RUN ;
**----- PROCESS macro invokation -----**;
%process (outn= 1,
    pop1= %str(Safety),
    pop2= %str(safety='Y' AND aete='Y'),
    ttl4= %str(Treatment-Emergent Adverse Events by &bslbl and Preferred Term),
    ttl5= %str(All Reported Adverse Events, including Occurrence) );

CONCLUSION

In new drug submissions to the Food & Drug Administration (FDA), adverse event tables play an important role in deciding whether a drug can move forward to the next trial phase and ultimately whether it can be approved. Producing these adverse event tables early in the study process can give statisticians and medical writers the time to address and possibly resolve any issues that the tables reveal in their early stages (such as incomplete standard coding or duplicate reporting of events).

The macros presented here will help speed up the process of producing these valuable tables. They count the number of Subjects who experienced a specific adverse event, in addition to the number of Occurrences of the event. In addition, the robustness that is built into these macros will especially help Clinical Research Organizations (CROs) who have to deal with a variety of clients, a variety of data collection patterns, and a variety of table requests. Even with the changing demands from client to client and study to study, these macros can be used to produce this set of standard set adverse events tables in a timely fashion.

VALIDATION

The macros shared in this paper have been fully tested and validated using SAS Version 9.1.3 software on Microsoft Windows ® XP-Professional platform.

REFERENCES

- SAS Macro Language, Reference, Ver. 9.1.

AUTHORS

Adel Fahmy has been a SAS user for 20 years. He worked as Independent Consultant, Sr. Statistical Programmer and Associate Director of Systems; at major pharmaceutical companies, clinical research organizations, and universities, teaching SAS to faculty and students. His special achievements include Generic Macros, Edit Checks, Database Design, Menu-Driven Systems and Optimization Techniques. Adel has a BS in Mathematics, a Graduate Diploma in Systems Design, a MS and a M.Phil. in Computer Science from Nottingham University, UK. Previous NESUG & PharmaSUG papers - 10.

Carol Matthews has been programming with SAS since 1990 and is currently the Senior Director of Clinical Programming for United BioSource Corporation, Biotechnology Solutions. Carol has been an adjunct faculty member in Philadelphia University’s SAS Programming certification program since 1998, has presented audio conferences through FDA News, and has recently published a book through SAS Press with co-author Brian Shilling titled "Validating Clinical Trial Data Reporting with SAS".

CONTACT

Your comments and questions are welcomed. The author can be contacted at:
Carol Matthews  EMS: Carol.Matthews@UnitedBioSource.com
Adel Fahmy, EMS: Adel.Fahmy@UnitedBioSource.com

COPYRIGHTS

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicate USA registration. Windows® is a registered trademark of Microsoft Inc. Other brand and product names are registered trademarks or trademarks of their respective companies.
APPENDIX 1
Sample Input Data

DM DATASET (Population Dataset)

<table>
<thead>
<tr>
<th>Variable</th>
<th>Type</th>
<th>Length</th>
<th>Format</th>
<th>Label</th>
</tr>
</thead>
<tbody>
<tr>
<td>USUBJID</td>
<td>Char</td>
<td>8</td>
<td>$8.</td>
<td>Unique Subject Identifier</td>
</tr>
<tr>
<td>ARMCD</td>
<td>Num</td>
<td>3</td>
<td>12.</td>
<td>Treatment Group (Num)</td>
</tr>
<tr>
<td>ARM</td>
<td>Char</td>
<td>8</td>
<td>$8.</td>
<td>Treatment Group (Char)</td>
</tr>
<tr>
<td>SAFETY</td>
<td>Char</td>
<td>1</td>
<td>$1.</td>
<td>Safety Population</td>
</tr>
</tbody>
</table>

AE DATASET (Adverse Events)

<table>
<thead>
<tr>
<th>Variable</th>
<th>Type</th>
<th>Length</th>
<th>Format</th>
<th>Label</th>
</tr>
</thead>
<tbody>
<tr>
<td>USUBJID</td>
<td>Char</td>
<td>8</td>
<td>$8.</td>
<td>Unique Subject Identifier</td>
</tr>
<tr>
<td>ARMCD</td>
<td>Num</td>
<td>3</td>
<td>12.</td>
<td>Treatment Group (Num)</td>
</tr>
<tr>
<td>ARM</td>
<td>Char</td>
<td>8</td>
<td>$8.</td>
<td>Treatment Group (Char)</td>
</tr>
<tr>
<td>SAFETY</td>
<td>Char</td>
<td>1</td>
<td>$1.</td>
<td>Safety Population</td>
</tr>
<tr>
<td>AESEQ</td>
<td>Num</td>
<td>8</td>
<td>12.</td>
<td>Sequence Number</td>
</tr>
<tr>
<td>AETERM</td>
<td>Char</td>
<td>100</td>
<td>$100.</td>
<td>Reported Term</td>
</tr>
<tr>
<td>AEBODSYS</td>
<td>Char</td>
<td>100</td>
<td>$100.</td>
<td>System Organ Class</td>
</tr>
<tr>
<td>AEDECOD</td>
<td>Char</td>
<td>100</td>
<td>$100.</td>
<td>Dictionary-Derived Term</td>
</tr>
<tr>
<td>AETE</td>
<td>Char</td>
<td>1</td>
<td>$1.</td>
<td>Treatment Emergent?</td>
</tr>
<tr>
<td>AESTDT</td>
<td>Num</td>
<td>8</td>
<td>DATE9.</td>
<td>Start Date of Adverse Event</td>
</tr>
<tr>
<td>AEENDT</td>
<td>Num</td>
<td>8</td>
<td>DATE9.</td>
<td>End Date of Adverse Event</td>
</tr>
<tr>
<td>AERELCD</td>
<td>Num</td>
<td>8</td>
<td>12.</td>
<td>Relation to Trial Drug</td>
</tr>
<tr>
<td>AEREL</td>
<td>Char</td>
<td>9</td>
<td>$9.</td>
<td>Relation to Trial Drug</td>
</tr>
<tr>
<td>AESEVCD</td>
<td>Num</td>
<td>8</td>
<td>12.</td>
<td>Severity/Intensity</td>
</tr>
<tr>
<td>AESEV</td>
<td>Char</td>
<td>9</td>
<td>$9.</td>
<td>Severity/Intensity</td>
</tr>
<tr>
<td>AESERCD</td>
<td>Num</td>
<td>8</td>
<td>12.</td>
<td>Serious Event</td>
</tr>
<tr>
<td>AESER</td>
<td>Char</td>
<td>1</td>
<td>$1.</td>
<td>Serious Event</td>
</tr>
<tr>
<td>AEOUTCD</td>
<td>Num</td>
<td>8</td>
<td>12.</td>
<td>Outcome</td>
</tr>
<tr>
<td>AEOUT</td>
<td>Char</td>
<td>24</td>
<td>$24.</td>
<td>Outcome</td>
</tr>
<tr>
<td>AEACNCD</td>
<td>Num</td>
<td>8</td>
<td>12.</td>
<td>Study Drug Action to Treat Event</td>
</tr>
<tr>
<td>AEACN</td>
<td>Char</td>
<td>9</td>
<td>$9.</td>
<td>Study Drug Action to Treat Event</td>
</tr>
</tbody>
</table>

APPENDIX 2
Sample Table Output

<table>
<thead>
<tr>
<th>System Organ Class/Preferred Term</th>
<th>Placebo n</th>
<th>N (%)</th>
<th>Drug A n</th>
<th>N (%)</th>
<th>Drug B n</th>
<th>N (%)</th>
<th>Total n</th>
<th>N (%)</th>
</tr>
</thead>
<tbody>
<tr>
<td>Population: Safety</td>
<td>136 (100.0)</td>
<td>131 (100.0)</td>
<td>131 (100.0)</td>
<td>398 (100.0)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Any Adverse Events</td>
<td>55 (32.4)</td>
<td>102 (42.7)</td>
<td>87 (41.2)</td>
<td>244 (38.7)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Cardiac disorders</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Atrial fibrillation</td>
<td>0 (3.3)</td>
<td>0 (1.5)</td>
<td>0 (0.8)</td>
<td>0 (0.8)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Palpitations</td>
<td>0 (1.1)</td>
<td>0 (0.8)</td>
<td>0 (0.8)</td>
<td>0 (0.8)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Gastrointestinal disorders</td>
<td>12 (8.1)</td>
<td>16 (12.9)</td>
<td>25 (13.0)</td>
<td>53 (10.1)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Diarrhoea</td>
<td>3 (2.2)</td>
<td>3 (2.3)</td>
<td>7 (4.6)</td>
<td>13 (3.0)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Constipation</td>
<td>5 (2.9)</td>
<td>2 (1.5)</td>
<td>6 (3.8)</td>
<td>11 (2.8)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Nausea</td>
<td>0 (0.8)</td>
<td>0 (0.8)</td>
<td>4 (3.1)</td>
<td>7 (1.8)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Abdominal pain</td>
<td>1 (0.7)</td>
<td>1 (0.8)</td>
<td>2 (1.5)</td>
<td>4 (1.0)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Stomach discomfort</td>
<td>0 (0.8)</td>
<td>0 (0.8)</td>
<td>3 (1.1)</td>
<td>7 (1.3)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Abdominal pain upper</td>
<td>3 (2.2)</td>
<td>1 (0.8)</td>
<td>1 (0.8)</td>
<td>5 (1.3)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Haemorrhoids</td>
<td>0 (0.8)</td>
<td>0 (0.8)</td>
<td>2 (0.8)</td>
<td>2 (0.5)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Vomiting</td>
<td>0 (0.8)</td>
<td>1 (0.8)</td>
<td>1 (0.8)</td>
<td>2 (0.5)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>